
1. Uvod

Pomemben pripomoček pri programeranju predstavlja možnost ponavljanja dela programa. Velikokrat se pri izvajanju programa pojavi potreba po ponavljanju določenih operacij. Ponavljanje se izvaja s posebnimi stavki, ki jih imenujemo zanke. Število ponovitev v zanki je lahko znano vnaprej, ali pa je odvisno od dobljenih rezultatov. Ukazi, ki omogočajo ponavljanje so dveh vrst:

· število ponovitev določenega dela je znano vnaprej

· zaporedje stavkov naj se ponavlja, dokler je izpolnjen določen pogoj.

Za izvajanje ponovitev se uporabljajo stavki:

· WHILE

· FOR

· REPEAT.

2. Stavek FOR

Stavek for se uporablja, ko vnaprej vemo število ponovitev. V algoritmih se pogosto pojavijo zanke, v katerih se neki celoštevilčni spremenljivki vrednost spreminja od določene začetne do končne vrednosti  s korakom po ena naprej. 

Sintaksa stavka FOR:

Izvajanje stavka for lahko opišemo takole:

a) for  ks := sm to zm do stavek ali

b) for ks :=  zm downto sm do stavek

ks je kontrolna spremenljivka tipa integer ali char, katere stavek ne sme spreminjati. Avtomatsko se postavi na spodnjo mejo (sm) in se povečuje (a) ali zmanjšuje (b) s korakom 1 do zgornje meje (zm), ali pa se zmajšuje s korakom ena navzdol od zgornje do spodnje meje. 

Zanko for lahko skoraj vedno zamenjamo z zanko while.

Sintaksa:

 
for  ks := sm to zm do stavek

je ekvivalentna sintaksi

ks:= sm;

while ks <=zm do begin

<stavek>


ks:= ks+1;

end;




Analogno velja za drugo varianto, kjer je stavek :

for  ks := zm downto sm do <stavek> ekvivalenten konstrukciji:

ks:= zm;

while ks >=sm do begin

<stavek>


ks:= ks-1;

end;

Če si predhodno prikazano programsko kodo podrobneje ogledamo, ugotovimo, da se v primeru, ko v prvem  koraku začetna vrednost spremenljivke (sm) ne ustreza pogoju zanke, stavek v zanki ne izvrši. 

3. Primeri programov z zanko for

· Primer 1: Izpiši cela števila na poljubnem intervalu med a in b!
Program napišemo tako, da uporabnika najprej vprašamo po prvem in nato po drugem številu. Program prebere števili in izpiše stavek v katerem pove, med katerimi števili bo izpisoval. Sledi zanka for, ki kontrolni spremenljivki i priredi vrednost a, in jo izpiše. Kontrolna spremenljivka se poveča za 1. Če vrednost kontrolne spremenljivke še zadošča pogoju izpiše naslednjo vrednost, če pa leta ne zadošča pogoju se stavek write (i,' '); ne izvede več.

program stevila;{Program izpise stevila na intervalu med a in b}

var

    a,b,i:integer;

begin

    write('Vpisi prvo stevilo: '); readln(a);

    write('Vpisi drugo stevilo: '); readln(b);

    writeln('Cela stevila med ', a,' in ', b, ' so:');

    for i:=a to b do

        write (i,' ');

end.

· Primer 2: Izpiši n večkratnikov danega  celega števila a.
Program prebere število, katerega večkratnike želimo izpisati in zahtevano število večkratnikov.  V zanki for množimo število a s kontrolno spremenljivko i in rezultat izpišemo. Zanka se ponavlja dokler ne pridemo do zahtevanega števila večkratnikov n.

program veckratniki;

var 

    n,stevilo,a,i : integer;

begin

    write ('Vpisi stevilo:  '); readln(a);

    write('Koliko veckratnikov naj napisem: '); readln(n);

    for i := 1 to n do begin

        stevilo := a * i;

        write (stevilo,'   ');

    end;

end.

· Primer 3: Izračunaj vsoto korenov določenega števila cifer.
V programu nastopata spremenljivki tipa integer:

n: število cifer, katerega korene želimo sešteti in

i:  kontrolna spremenljivka. 

V program najprej vpišemo koliko korenov števil od 1 do n želimo sešteti. Vsota korenov števil je v začetku nič. Sledi stavek for v katerem seštevamo korene. Zanka for se konča, ko je i>n. Vsota korenov je tipa real in jo izpišemo na dve decimalki natančno.  

program vsote_korenov;

{program izpiše vsoto korenov določenega stevila naravnih stevil}

var

  n,i: integer;

  vsota: real;

begin

  vsota := 0;

  writeln('Vpisi koliko stevil naj vzamem: '); readln(n);

  for i := 1 to n do

  vsota := vsota + sqrt(i);

  writeln('Vsota prvih ',n, 'stevil je ', vsota:1:2);

end.

· Primer 4: Izpiši n členov rekurzivno podanega zaporedja: 
a1=x, a2=y, an=(3an-1 + 4an-2)

Pravilo imamo podano za tretji člen zaporedja, zato mora program prebrati prva dva člena zaporedja in število izpisanih členov zaporedja. Zanko for uporabimo za izračun tretjega in vseh ostalih členov. V zanki izračunamo člen an (a3) in predhodnima členoma (a1, a2) priredimo nove vrednosti. V vsakem koraku izpišemo vrednost člena zaporedja.

program zaporedje;

{Program izpiše n členov rekurzivno podanega zaporedja}

var

   a1,a2,a3,n,i: integer;

begin

   write('Vpisi prvi clen zaporedja: '); readln(a1);

   write('Vpisi drugi clen zaporedja: '); readln(a2);

   write('Vpisi koliko clenov zaporedja zelis izpisati:'); readln(n);

   writeln('a1= ',a1 );

   writeln('a2= ',a2 );

   for i:=3 to n do begin

   a3:= (3*a2+ 4*a1);

   a1:= a2;

   a2:= a3;

   writeln('a',i, '= ',a3);

   end;

end.

· Primer 5 : Napiši program, ki zamenja vse samoglasnike z znakom #.
Nastopajo trije tipi spremenljivk: poljubno besedilo tipa string, kontrolna spremenljivka tipa integer in željeni znak za zamenjavo samoglasnika tipa char. Najprej vpišemo besedilo. Sledi zanka for v kateri uporabimo funkcijo length, ki vrne dolžino stringa. Vsak znak v stringu besedilo preverimo stavkom if, če je samoglasnik. V primeru da je ga zamenjamo z znakom #, v nasprotnem primeru pa izpišemo znak na tem mestu. Zanka for se ponavlja, dokler ne pridemo do konca besedila.

program samoglasniki;

{Program zamenja vse samoglasnike v danem nizu z znakom #}

var

  besedilo: string;

  i: integer;

  znak: char;

begin

    write('Vpisi besedilo: ');  readln(besedilo);

    for i:= 1 to Length(besedilo) do begin

        znak := (besedilo[i]);

        if (znak ='a') or (znak='e') or (znak ='i') or (znak='o') or

        (znak ='u') or (znak ='A') or (znak='E') or (znak ='I') or 

        (znak='O') or (znak ='U') then  

            znak:='#'

        else

            znak:= znak;

        write (znak); 

    end;

end.

V zanki for kot kontrolno spremenljivko lahko uporabimo  tip "char".

· Primer 6: Izpiši obrnjeno abecedo velikih črk in abecedo malih črk.
Program vsebuje samo eno spremenljivko tipa char. V prvi zanki for izpisujemo velike črke v obrnjenem vrstnem redu, v naslednji pa male črke od a do z.

program abeceda;

{Program izpise abecedo malih cek in obrnjeno abecedo velikih crk}

var

  znak: char;

begin

    for znak:= 'Z' downto 'A' do write(znak,' ');

    writeln;

    for znak:= 'a' to 'z' do write(znak,' ');

end.

Pogosto uporabimo gnezdenje zank.

· Primer 7: Izpiši vsa števila do tisoč z zahtevano vsoto števk.
Z zaslona program prebere zahtevano vsoto števk. Za vsako naravno število od ena do tisoč (zanka for), se v zanki while zaporedoma računa celoštevilsko deljenje z deset in njegov ostanek, ki predstavlja zadnjo števko v deljenem številu. Zanka se prekine, ko je rezultet celoštevilskega deljenja enak nič.

program vsota_stevk;

{Izpise stevila z zahtevano vsoto stevk}

var

    i,j: integer;

    vsota,suma: integer;

begin

    writeln ('Izpisal bom naravna stevila do tisoč z zahtevano vsoto stevk.');

    write('Vsota stevk naj bo: '); readln(vsota);

    writeln ('Naravna stevila z vsoto stevk  ',vsota, '  so:  ');

    for i := 1 to 1000 do begin

        j := i;

        suma := 0;

        while j > 0 do begin

            suma := suma + (j mod 10);

            j := j div 10;

        end;

        if suma = vsota then 

              write (i:5);

    end;

end.    

· Primer 8: Nariši smreko
Program je sestavljen iz štirih delov: risanje krošnje, risanje debla in podrasti ter risanje tal. Krošnja je trikotnik  visok 12 vrstic.  V prvih dveh vrsticah je po ena zvezdica, tretja in četrta sta sestavljeni iz treh zvezdic, v naslednijh pa imamo še po dve zvezdici več. Postopek za risnje krošnje je sledeč : 

za i =1,2,3…6 dvakrat: na sredini vrste nariši 2i –1 zvezdic.

To realiziramo s stavkom for.

Risanje debla je enostavnejše. Na sredini vsake vrstice izpišemo le en znak, črko H. Spodnje vrstice pa še okrasimo s podrastjem.

program smreka; {risanje drevesa}

var

   i, j, k:integer;

begin

    for i:=1 to 3 do writeln; {prazne vrstice}

    for i:=1 to 6 do begin {risanje krosnje}

        for j:=1 to 2 do begin 

            write (' ':20-i);

            for k:=1 to 2*i-1 do write('*'); {na sredini 2i-1 zvezdic}

            writeln

        end;

    end;

    for i:=1 to 4 do writeln('H':20); {deblo}

        writeln(' ':13, 'v VV vvH vv v');{podrast}

    write(' ':12);

    for i:=1 to 15 do write ('X'); {tla}

    writeln;

end. 

Vaje

1. Napiši program, ki zmnoži števila na intervalu med a in b.

2. Napiši program, ki izpiše vse delitelje danega števila!

3. Izpiši kvadrate naravnih števil od 1 do n. n je naravno število med 1 in 100.

4. Napiši program, ki bo s pomočjo Ohmovega zakona izračunal tok čez upornik, če napetost spreminjaš po 1V. V programu določiš začetno in končno napetost, ter njegovo upornost.

5. Program naj za dani pospešek izračuna hitrost in opravljeno pot telesa. Časovni interval naj narašča od 1s do vrednosti, ki jo določi uporabnik.

6. Napiši program, ki bo za poljubno število krogov  prebral polmer in izračunal ploščino in obseg vsakega kroga.

7. Napiši program, ki izpiše znake med prebranima znakoma in njihove ordinalne vrednosti.

8. Nariši  pravokotnik velikosti a*b znakov!

9. Napiši program, ki za dano naravno število n izpiše vsa naravna števila med 1 in 1000 z vsoto števk enako n.

10. Izpiši n členov rekurzivno podanega zaporedja: 

a0=x, a1=y, a3=z, an=(an-1 + 4an-2) *an-3
11. Kote v stopinjah pretvori v radiane. Preberi, koliko kotov želiš pretvoriti. Preberi kote v stopinjah in izpiši njihove vrednosti v radianih.

12. Napiši program, ki prebere besedilo in v njem podvoji črke a, z, k, o.

13. Program naj na zaslon nariše:

